

# POZNÁMKA K NUMERICKÉ PŘESNOSTI MODIFIKOVANÉHO PICKETTOVA MODELU MEMRISTORU

Dalibor Bielek<sup>1,2</sup>

<sup>1</sup>Katedra elektrotechniky; Fakulta vojenských technologií, Univerzita obrany, Brno, dalibor.bielek@unob.cz

<sup>2</sup>Ústav mikroelektroniky; Fakulta elektrotechniky a komunikačních technologií VUT v Brně, bielek@feec.vutbr.cz

## Abstrakt

V článku jsou navrženy modifikace matematického modelu TiO<sub>2</sub> memristoru, který vychází z aproximací Simmonsových rovnic tunelového efektu v strukturách MIM (Metal-Insulator-Metal). Cílem modifikací je dosažení korektního numerického chování modelu v programech typu SPICE.

Klíčová slova: memristor, Pickettův model, SPICE.

## Abstract

Modifications of the mathematical model of the TiO<sub>2</sub> memristor, based on the approximations of Simmon's equations of tunnel effects in MIM (Metal-Insulator-Metal) structures are proposed. The aim of the modifications is to provide correct numerical behavior of the model in SPICE-family programs.

Keywords: memristor, Pickett model, SPICE.

## 1 Úvod

Pickettův model [1] patří k důležitým fyzikálně-fenomenologickým modelům tzv. TiO<sub>2</sub> memristoru. Jeho portová rovnice, tj. rovnice mezi napětím a proudem, vychází ze Simmonsových rovnic tunelového efektu v přechodech kovizolant [2]. Naproti tomu stavová rovnice modelu, tj. diferenciální rovnice pro šířku tzv. bariéry  $w$ , je čistě fenomenologická. V [3] je ukázáno, že nedůsledná aplikace Simmonsova modelu v [1] je příčinou nekorektního chování simulačního programu v určitých mezních stavech, do nichž se může dostat iterační algoritmus simulátoru. Proto bylo navrženo několik typů aproximací branové rovnice, z nichž poslední je popsána v [4], které daný problém odstraňují. Nicméně dalším zdrojem potenciálních numerických problémů je vlastní stavová rovnice. Dynamika dějů v TiO<sub>2</sub> memristoru je natolik výrazná, že derivace stavové proměnné je modelována funkcemi typu  $\exp(\exp(\cdot))$ . To s sebou nese reálné nebezpečí přetečení/podtečení numerických dat, neboť numerické limity v programech typu SPICE jsou dány formátem „double“ (Double-Precision Binary Floating-Point).

Článek navazuje na práci [4], která tuto problematiku řeší jen okrajově. Navrhovaná opatření jsou optimalizována pro simulační program Cadence PSpice 16, ale přiměřeně je možné jich využít i u dalších programů třídy SPICE.

## 2 Aproximační Pickettův model

Vztah mezi napětím v tunelového přechodu a jím protékajícím proudem  $i$  je navržen v [4] ve formě aproximace

$$i = \text{sgn}(v)k_1k_2^w \left[ \sinh((k_3 + k_4w)|v|) + k_5(\exp(k_6|v|) - 1) \right] \quad (1)$$

kde  $\text{sgn}(\cdot)$  je znaménková funkce,  $w$  je šířka bariéry, a  $k_1$  až  $k_5$  jsou parametry optimalizované tak, aby model (1) věrně popisoval chování přechodu v napěťovém rozsahu  $|v| < 1,2$  V. Číselné hodnoty parametrů jsou uvedeny v příloze.

Součástí modelu je přídavný sériový odpor  $R_S = 232 \Omega$ , který reprezentuje prakticky konstantní odpor vysoce vodivé dopované vrstvy TiO<sub>2-x</sub>. Podrobnosti jsou uvedeny v [4].

Originální stavová rovnice pro šířku bariéry  $w$  z [1] má tvar

$$\frac{dw}{dt} = \begin{cases} +f_{\text{off}} \sinh\left(\frac{|i|}{i_{\text{off}}}\right) \exp\left(-\frac{w}{w_c}\right) \exp\left(-e^{-\frac{w-a_{\text{off}}-|i|}{w_c b}}\right) & \dots i \geq 0 \\ -f_{\text{on}} \sinh\left(\frac{|i|}{i_{\text{on}}}\right) \exp\left(-\frac{w}{w_c}\right) \exp\left(-e^{-\frac{a_{\text{on}}-w-|i|}{w_c b}}\right) & \dots i < 0 \end{cases} \quad (2)$$

Hodnoty příslušných parametrů z (2) lze opět vyčíst ze zdrojových kódů v příloze.

V [4] je ukázáno, že problémem rovnic (1) a (2) je přetékání výsledků operací typu  $\exp(\cdot)$  během transientní analýzy nad povolenou horní hranici numerické reprezentace dat. Tento problém byl řešen dvěma způsoby, jednak tangenciální aproximací exponenciální funkce a z ní odvozené funkce hyperbolického sinu, jednak jednoduchou úpravou stavové rovnice (2). Tyto modifikace však neřeší numerické problémy uvedených modelů beze zbytku.

V článku je proveden návrh komplexního řešení s ohledem na numerická omezení programů třídy SPICE.

## 3 Numerická omezení simulačního programu Cadence PSpice 16.6

Numerická omezení vycházejí z formátu dat typu „Double-Precision Binary Floating-Point“, který umožňuje reprezentaci čísel v dynamickém rozsahu od  $2^{-1022}$  do  $2^{+1023}$ , což je přibližně  $10^{-308}$  až  $10^{+308}$  nebo  $e^{-709}$  až  $e^{+709}$ , kde  $e$  je základ přirozených logaritmů. Specifickým omezením programů SPICE je navíc limitace velikosti počítaných napětí [V] a proudů [A] na interval od  $10^{-30}$  do  $10^{+10}$ . Konkrétně to znamená, že pokud by vypočtená hodnota napětí měla být  $|v| < 10^{-30}$  V, SPICE ji změní na nulu. Naopak případ  $|v| > 10^{+10}$  V ukončí simulaci chybovým hlášením „Convergence problem in bias point calculation“. Dále je třeba si uvědomit, uvedené konvergenční problémy se objeví při libovolné iteraci, kdy obvodová veličina překročí limit  $10^{+10}$ , i když může být hodně vzdálená od hodnoty, k níž by měl výpočet směřovat.

Přetečení, resp. podtečení proměnných mimo dynamický rozsah typu „double“ není důvodem k přerušení výpočtu, avšak proměnná je změněna, což vede ke generování skrytých numerických chyb. V Cadence PSpice 16.6 k tomuto jevu dochází již „postupně“, přibližuje-li se výsledek matematické operace k danému limitu. Definujme například zdroj napětí

```
Ex x 0 value={ (2^1017) / (2^1016) } .
```

Pak napětí uzlu x je vypočteno správně:

```
(x) 2.0000
```

Další přiblížení k numerickému limitu  $2^{1023}$  již vede k chybnému výsledku:

```
Ex x 0 value={ (2^1018) / (2^1017) } ,
```

```
(x) 1.0718
```

#### Kód

```
.param b1=706 b2=700 b3={b1-b2} b4=716 b5=710
E1 1 0 value={exp(b1)/exp(b2)}
E2 2 0 value={exp(b1-b2)}
E3 3 0 value={exp(b3)}
E4 4 0 value={exp(b4)/exp(b5)}
E5 5 0 value={10^(-29)}
E6 6 0 value={10^(-30)}
E7 7 0 value={-(10^(-30))}
```

vygeneruje uzlová napětí, z nichž napětí uzlů 1, 4, 6 a 7 jsou nesprávná:

```
(1) 148.4100 (2) 403.4300 (3) 403.4300
(4) 1.0000 (5) 10.000E-30 (6) 0.0000 (7) 0.0000
```

Tento příklad potvrzuje výše uvedená tvrzení, konkrétně:

- blíží-li se výsledek dílčí matematické operace k teoretickému limitu  $e^{709}$ , dojde k jeho chybnému vyčíslení;
- překročí-li výsledek dílčí matematické operace limit  $e^{709}$ , pak je omezen na fixní hodnotu;
- je-li absolutní velikost napětí menší nebo rovna limitu  $10^{-30}$ , pak je toto napětí interpretováno jako nulové.

U modifikovaného Pickettova modelu (1) a (2), který obsahuje exponenciální výrazy a dokonce výrazy typu  $\exp(\exp(\dots))$ , je tedy potenciální nebezpečí výskytu jak skrytých numerických chyb, tak i vážných konvergenčních problémů v průběhu iteračního algoritmu transientní analýzy. Například při proudu  $i = -7$  mA a parametru  $i_{on} = 8,9$   $\mu$ A vychází  $i/i_{on} = -786,5$ , takže u příslušné exponenciální funkce  $\exp(-i/i_{on})$ , která je součástí funkce  $\sinh(i/i_{on})$  v rovnici (2) dojde k jejímu chybnému vyhodnocení.

Podobné numerické problémy mohou vznikat u portové rovnice (1). V [3] je demonstrováno, jak při iteračním řešení jednoduchého obvodu s memristorem může nastat situace, kdy simulační program začne sledovat chybné, fyzikálně neexistující řešení branové rovnice. Příčinou je značný odklon velikosti napětí v dané iteraci od hodnoty hledané. Zabloudí-li iterační algoritmus do oblasti extrémně vysokých napětí, může to v krajním případě vyvolat i přetečení uzlových napětí

či branových proudů nad mez  $10^{10}$  a následně konvergenční problémy.

## 4 Modifikace numerického modelu

Definujme funkci

$$\lns(x) = \begin{cases} x - \ln(x) + \ln(1 - e^{-2x}) & \text{pro } x > 0 \\ 0 & \text{pro } x \leq 0 \end{cases} \quad (3)$$

S jejím využitím je možno stavovou rovnici (2) přepsat do tvaru

$i \geq 0$ :

$$n \frac{dw}{dt} = \exp \left[ \ln(nf_{off}) + \lns \left( \frac{i}{i_{off}} \right) - \frac{w}{w_c} - \exp \left( \frac{w - a_{off}}{w_c} - \frac{i}{b} \right) \right] \frac{i}{i_{off}} \quad (4)$$

$i < 0$ :

$$n \frac{dw}{dt} = \exp \left[ \ln(nf_{on}) + \lns \left( -\frac{i}{i_{on}} \right) - \frac{w}{w_c} - \exp \left( \frac{a_{on} - w}{w_c} + \frac{i}{b} \right) \right] \frac{i}{i_{on}}$$

Symbolem  $n$  je označen tzv. normovací faktor, který bude diskutován později.


Smyslem této úpravy je vyhodnocovat pravé strany diferenciální rovnice (2) prostřednictvím jejich logaritmů, čímž se omezí možnost přetečení exponenciálních funkcí. Na pravé straně (4) sice zůstávají další exponenciální funkce, jejich argumenty se však vzhledem k číselným hodnotám daných parametrů pohybují během transientní analýzy hluboko pod kritickou hranici 709.

Použijeme-li stejnou strategii na portovou rovnici (1), můžeme ji upravit na tvar

$$i = \left\{ \exp \left[ w \ln(k_2) + \ln(k_3 + k_4 w) \right] + \lns \left( (k_3 + k_4 w) |v| \right) \right\} + k_5 k_6 \exp \left[ w \ln(k_2) + k_6 |v| / 2 + \lns \left( k_6 |v| / 2 \right) \right] k_1 v \quad (5)$$

Kód podobvodu SPICE, příslušející upravenému modelu (4), (5), je uveden v příloze. Vyplývá z něj, že pro  $n = 1$  (normovací faktor z (4)) by k integraci diferenciální rovnice (4) byl použit kapacitor o kapacitě 1 nF, převádějící proud, odpovídající pravé straně (4), na napětí kapacitoru, tj. integrál proudu, násobený inverzní kapacitou, tedy faktorem  $10^9$ . Napětí na kapacitoru by tedy číselně odpovídalo šířce bariéry  $w$  v nanometrech. Počítačové simulace však ukazují, že pravá strana (4), která odpovídá proudu zdroje GI, nabývá velmi nízkých hodnot (desítky nA). V zájmu zvýšení přesnosti numerických výpočtů je vhodné tento proud zvětšit, avšak při současném zachování velikosti napětí na kapacitoru. Toho lze snadno dosáhnout současným zvětšením jak proudu, tak i integrační kapacity  $n$  krát. Pro  $n = 1e7$  pak vychází proud, nabíjející kapacitor, stovky mA.

**Obrázek 1** ukazuje výsledky transientní analýzy známé simulační úlohy, popsané v originálním článku [1]: sériová kombinace TiO<sub>2</sub> memristoru a rezistoru o odporu 2,4 k $\Omega$  je buzena nesymetrickým trojúhelníkovým napětím (na obr. 1 označeným symbolem V(in)). Odpovídající hysterezní smyčka v rovině napětí-proud je ve velmi dobré shodě s experimentálními výsledky z [1].



**Obr. 1.** Výsledky transientní analýzy obvodu s  $\text{TiO}_2$  memristorem buzeného ze zdroje nesymetrického trojúhelníkového napětí. Nahoře: hysterezní smyčka v rovině napětí-proud, dole: časové průběhy budícího napětí (modře) a proudu memristorem (červeně).

## Poděkování

Článek vznikl za podpory COST Action IC1401 financovaného MŠMT pod projektem číslo LD15033 a za podpory projektu Specifického výzkumu K217 UO v Brně.

## Literatura

- [1] Abdalla, H., Pickett, M. D. SPICE modeling of memristors. In *Proc. 2011 IEEE Int. Symp. Circ. Syst. (ISCAS)*, 2011, p. 1832–1835.
- [2] Simmons, J. G. Generalized formula for the electric tunnel effect between similar electrodes separated by a thin insulating film. *Journal of Applied Physics*, 1963, vol. 34, no. 6, p. 1793-1803.
- [3] Kolka, Z., Biolek, D., Biolková, V. Improved model of  $\text{TiO}_2$  memristor. *Radioengineering*, 2015, vol. 24, no. 2, p. 378-383.
- [4] Biolek, D., Kolka, Z., Biolková, V., Biolek, Z., Nosek, J. Modelování a simulace rozsáhlých memristivních sítí. *Slaboproudý obzor*, 2015, vol. 71, no. 4, p. 1-6.

## Příloha:

### PSPICE zdrojový kód pro upravený Pickettův model

```
.SUBCKT MPmodel_new A B params: wini=1.2
*Modified Pickett model
.param foff=3.5u ioff=115u aoff=1.2 fon=40u
+ ion=8.9u aon=1.8 b=500u wc=107m
.param k1=11.3153 k2=44.6944u k3=1.34192
+ k4=3.0364 k5=11.4919n k6=24.1384 Rs=232.047
*lns(x)=ln(sinh(x)/x)
.func lns(x)={if(x<=0,0,x-log(x)+log(1-exp(-2*x)))}
*Port Equation
.func jg(v,w)={ (exp(w*log(k2)+log(k3+k4*w)+
+lns((k3+k4*w)*v))+k5*k6*exp(w*log(k2)+
+k6*v/2+lns(k6*v/2)))*k1*v}
GM A int value={if(v(A,int)>=0,jg(v(A,int),v(w)),
+-jg(-v(A,int),v(w)))}
Rp A int 1T
Rs int B {Rs}
*State Equation
.param scale=1e7
.func logderioff(im,w)={log(scale*foff)+
+lns(im/ioff)-w/wc-exp((w-aoff)/wc-im/b)}
.func logderion(im,w)={log(scale*fon)+
+lns(-im/ion)-w/wc-exp((aon-w)/wc+im/b)}
.func coff(im,w)={exp(logderioff(im,w))*im/ioff}
.func con(im,w)={exp(logderion(im,w))*im/ion}
GI 0 w value={if(V(int,B)>=0,
+coff(V(int,B)/Rs,v(w)),con(V(int,B)/Rs,v(w)))}
CI w 0 {ln*scale}
RI w 0 1T
.IC v(w)={wini}
.ENDS MPmodel_new
```